

APPENDIX 1

LISTING OF THE COMPUTER PROGRAM GCCONT

```

00100 REM      THIS PROGRAM, GCCONT, FINDS PREVALENCE AND INCIDENCE OF GONORRHEA
00120 REM FOR SEVERAL CONTROL SITUATIONS. TO MAKE A RUN, ENTER PARAMETERS WHEN
00140 REM THEY ARE REQUESTED.
00180 REM
00200 REM      THE MODEL EQUATIONS ARE COMPUTED IN THE 'RATE' SUBROUTINE AND THE
00220 REM 'CONTROL' SUBROUTINE, USING THE CONTACT MATRIX WHICH THE 'PATTERN'
00240 REM SUBROUTINE CALCULATED.
00260 REM
00280 REM      THE PROGRAM IS ORGANIZED AS FOLLOWS:
00300 REM AFTER WE ENTER DATA, THE PROGRAM MAKES PRELIMINARY CALCULATIONS WHICH
00320 REM INCLUDE CALLING THE PATTERN SUBROUTINE TO FORM THE CONTACT MATRIX.
00340 REM THEN THE MAIN PROGRAM LOOP BEGINS: DURING A RUN THE LOOP VARIABLE R1
00360 REM STEPS THROUGH THE FOLLOWING VALUES:
00380 REM      R1=1 FOR THE UNCONTROLLED CASE
00400 REM      R1=2 FOR POPULATION SCREENING (WOMEN)
00420 REM      R1=3 FOR POP SCREENING PLUS SUPPLEMENTARY POP SCREENING
00440 REM      R1=4 FOR POP SCREENING PLUS SUPPLEMENTARY INFECTEE TRACING
00460 REM      R1=5 FOR POP SCREENING PLUS SUPPLEMENTARY INFECTOR TRACING
00480 REM
00500 REM FOR EACH CASE, THE MAIN PROGRAM CALLS THE 'SOLVER' SUBROUTINE WHICH
00520 REM USES ITERATION PROCEDURES TO FIND PREVALENCES. THE MAIN PROGRAM ALSO
00540 REM MAKES SUMMARY CALCULATIONS FOR EACH CASE AND PRINTS OUT A TABLE OF
00560 REM RESULTS. THE 'SOLVER' SUBROUTINE ORGANIZES THE ITERATION PROCESS.
00580 REM THE 'RATE' SUBROUTINE ACTUALLY COMPUTES THE DIRECTION FIELD V(I,1):
00600 REM IT CALLS THE 'CONTROL' SUBROUTINE TO FIND THE CONTROL RATE PORTION.
00620 REM RATE ALSO STORES THE PREVIOUS VALUE OF THE DIRECTION FIELD IN V(I,2)
00640 REM AND COMPUTES THE DIFFERENCE V(I,3)=V(I,1)-V(I,2). (OTHER NUMERICAL
00660 REM METHODS MIGHT REQUIRE MORE BACK VALUES WHICH COULD BE STORED IN
00680 REM OTHER COLUMNS OF V.) 'SOLVER' USES AN EULER STEP METHOD TO BEGIN,
00700 REM THEN CALLS THE 'NEWTON' SUBROUTINE FOR NEWTON'S METHOD. THE MATRIX OF
00720 REM PARTIAL DERIVATIVES IS FOUND BY THE 'FDERIV' SUBROUTINE. WHEN THE
00740 REM RMS RATE IS SMALL ENOUGH, 'SOLVER' STOPS THE ITERATION PROCESS, AND
00760 REM THE MAIN PROGRAM SUMMARIZES RESULTS, THEN STEPS TO THE NEXT CONTROL
00780 REM CASE.
00800 REM
00820 REM DIMENSIONS OF VECTORS AND MATRICES:
00840 REM IF THE NUMBER OF GROUPS G IS NOT 4, REPLACE 4 BY G IN DIM STATEMENTS
00860 REM AND ALSO MODIFY PRINT LINES IN WHICH WE LABEL 4 GROUPS EXPLICITLY.
00880 DIM N(8),Y(8),D(8),K(8),P(8)
00900 DIM R(8),Z(8),W(8),U(8),E(8)
00920 DIM M(8,8),B(8,8),T(8,8),L(8,8)

```

```

00940 DIM A(8,8),V(8,3),F(8,1),X(8,1),H(8,1)
00960 DIM S(5,8),Q(11,8),O(1,8)
00980 REM
01000 REM - - - - - ENTER DATA - - - - -
01020 REM
01040 REM INDEXING:  WOMEN ODD, MEN EVEN
01060 REM ORDERING:  GP1 WOMEN, GP2 MEN, GP3 WOMEN, GP4 MEN
01080 REM G=NUMBER OF GROUPS
01100 G=8
01120 PRINT'INPUT G1=SELF INTERACTION FRACTION'
01140 INPUT G1
01200 REM G2 SELECTS PRINTOUT: G2=0 FOR SUMMARY ONLY:  G2=1 FOR ITERATION INFO
01220 G2=0
01240 REM N(I)=RELATIVE SIZE OF GROUP I, IN ANY UNITS
01260 PRINT 'INPUT GROUP FRACTIONS N(I)'
01280 MAT INPUT N
01300 PRINT'INPUT ACTIVITY RATIOS OF GROUPS:WOMEN AND MEN'
01320 INPUT R8,R9
01340 REM Y(I)=INITIAL GUESS OF PREVALENCE, THE INFECTIVE FRACTION OF GROUP I
01360 FOR I=1 TO G STEP 1
01380 READ Y(I)
01400 NEXT I
01420 DATA .5,.5,.1,.1,.1,.1,.1,.1
01440 REM D(I)=DURATION OF DISEASE (AVERAGE) FOR GROUP I, IN ANY UNITS (MONTHS)
01460 PRINT 'INPUT DURATIONS D(I)'
01480 MAT INPUT D
01500 PRINT'INPUT PROB OF TRANS BY INF MAN/PROB OF TRANS BY INF WOMAN'
01520 INPUT R7
01540 REM K(I)=NUMBER OF EFFECTIVE CONTACTS OF EACH GROUP I INFECTIVE
01560 REM DURING THE INFECTIOUS PERIOD.
01580 PRINT'INPUT K(1)'
01600 INPUT K(1)
01620 K(2)=K(1)*D(2)/D(1)*R7*(N(1)+N(5)+(N(3)+N(7))/R8)/(N(2)+N(6)+(N(4)+N(8))/R9)
01640 K(3)=K(1)*D(3)/D(1)/R8
01650 K(4)=K(2)*D(4)/D(2)/R9
01660 K(5)=K(1)*D(5)/D(1)
01670 K(6)=K(2)*D(6)/D(2)
01680 K(7)=K(1)*D(7)/D(1)/R8
01690 K(8)=K(2)*D(8)/D(2)/R9
01780 REM - - - - -
01800 REM PRELIMINARY CALCULATIONS
01820 MAT A=ZER(G,G)
01840 MAT F=ZER(G,1)
01860 MAT X=ZER(G,1)
01880 MAT H=ZER(G,1)
01900 MAT V=ZER(G,3)
01902 PRINT
01904 PRINT '#####  RUN NUMBER  #####'
01918 PRINT
01920 PRINT ' ', ' GONORRHEA CONTROL MODEL'
01940 PRINT ' ', ' -----'
01980 REM
02000 REM GO BUILD CONTACT MATRIX B
02020 REM
02040 GOSUB 8680
02060 REM
02080 REM ***** MAIN PROGRAM LOOP *****
02100 REM
02120 REM R1 INDEXES THE CONTROL METHOD: R1=1 BEFORE THE CONTROLS ARE USED
02140 R1=1

```

```

02160 REM GO TO THE SOLVER SUBROUTINE TO FIND THE SOLUTION
02180 GOSUB 4560
02200 IF G2=0 THEN 2280
02220 PRINT Y(1),Y(2),Y(3),Y(4),W5
02240 PRINT 'SOLUTION APPROACHED IS ON ABOVE LINE', ' ', 'RMS RATE'
02280 IF R1>1 THEN 2680
02300 REM SAVE VALUES FROM ORIGINAL CASE, BEFORE APPLYING CONTROLS
02320 S2=0
02340 S3=0
02360 S4=0
02380 S5=0
02400 FOR I=1 TO G STEP 1
02420 S(1,I)=Y(I)
02440 S(2,I)=Y(I)*N(I)
02460 S(3,I)=D(I)
02480 S(4,I)=S(2,I)/D(I)
02500 Z(I)=S(4,I)
02520 NEXT I
02540 FOR I=1 TO G-1 STEP 2
02560 S2=S2+Z(I+1)
02580 S3=S3+Z(I)
02600 S4=S4+S(2,I+1)
02620 S5=S5+S(2,I)
02640 NEXT I
02660 REM STORE VALUES FOR PRINTING IN MATRIX Q
02680 Q2=0
02700 Q3=0
02720 Q4=0
02740 Q5=0
02760 FOR I=1 TO G STEP 1
02780 Q(1,I)=Y(I)
02800 Q(2,I)=(Q(1,I)-S(1,I))*100/S(1,I)
02820 Q(3,I)=Y(I)*N(I)
02840 Q(4,I)=(Q(3,I)-S(2,I))*100/S(2,I)
02860 Q(5,I)=D(I)
02880 Q(6,I)=1/(1/D(I)+C*R(I)/(Y(I)*N(I))+.1*C*P(I)/(Y(I)*N(I)))
02900 Q(7,I)=Q(3,I)/Q(6,I)
02920 Q(8,I)=(Q(7,I)-S(4,I))*100/S(4,I)
02940 Q(9,I)=C*R(I)+.1*C*P(I)
02960 Q(10,I)=N(I)
02980 Q(11,I)=K(I)
03000 NEXT I
03020 FOR I=1 TO G-1 STEP 2
03040 Q2=Q2+Q(7,I+1)
03060 Q3=Q3+Q(7,I)
03080 Q4=Q4+Q(3,I+1)
03100 Q5=Q5+Q(3,I)
03120 NEXT I
03160 ON R1 GOTO 3180,3220,3230,3260,3320
03180 PRINT '***** RESULTS WITH NO EXTRA CURE RATE APPLIED *****'
03200 GOTO 3340
03220 PRINT '***** RESULTS FOR POPULATION SCREENING *****'
03221 PRINT 'THE NUMBER OF WOMEN DISCOVERED BY POPULATION SCREENING IS'
03222 PRINT 'SET EQUAL TO 10% OF THE INCIDENCE AND THE EFFECTIVE CONTACT'
03223 PRINT 'NUMBER IS CHOSEN SO THAT THIS POPULATION SCREENING CAUSES'
03224 PRINT 'APPROXIMATELY A 20% DECREASE IN INCIDENCE IN MEN.'
03225 GO TO 3340
03230 PRINT '*****RESULTS FOR SUPPLEMENTARY POPULATION SCREENING *****'
03235 GO TO 3340
03260 PRINT '***** RESULTS FOR SUPPLEMENTARY INFECTEE TRACING *****'

```

```

03280 PRINT 'TRACE WOMEN REPORTED AS CONTACTEES OF DIAGNOSED MEN'
03300 GOTO 3340
03320 PRINT '***** RESULTS FOR SUPPLEMENTARY INFECTOR TRACING *****'
03330 PRINT 'TRACE WOMEN REPORTED AS INFECTORS OF DIAGNOSED MEN'
03340 PRINT ' ', 'GP1 WOMEN', 'GP2 MEN', 'GP3 WOMEN', 'GP4 MEN'
03440 FOR I1=1 TO 11 STEP 1
03460 FOR I=1 TO G STEP 1
03480 O(1,I)=Q(11,I)
03500 NEXT I
03540 ON I1 GOTO 3560,3600,3640,3680,3720,3760,3800,3840,3880,3920,3960
03560 PRINT 'PREVALENCE,FRAC':
03580 GOTO 3980
03600 PRINT 'CHANGE (%) ':
03620 GOTO 3980
03640 PRINT 'PREVALENCE, POP':
03660 GOTO 3980
03680 PRINT 'CHANGE (%) ':
03700 GOTO 3980
03720 PRINT 'DURATION,ORIG ':
03740 GOTO 3980
03760 PRINT 'DURATION, NEW ':
03780 GOTO 3980
03800 PRINT 'INCIDENCE,POP':
03820 GOTO 3980
03840 PRINT 'CHANGE (%) ':
03860 GOTO 3980
03880 PRINT 'EXTRA CURE RATE':
03900 GOTO 3980
03920 IF R1>1 THEN 4040
03930 PRINT 'POPULAT. RATIOS':
03940 GOTO 3980
03960 PRINT 'CONTACT NUMBERS':
03980 PRINT O(1,1),O(1,2),O(1,3),O(1,4)
03990 PRINT ' GPS 5,6,7,8',O(1,5),O(1,6),O(1,7),O(1,8)
04000 NEXT I1
04040 PRINT ' WOMEN,POP PREVALENCE =',Q5,'CHANGE (%)= ',(Q5-S5)*100/S5
04060 PRINT ' MEN, POP PREVALENCE =',Q4,'CHANGE (%)= ',(Q4-S4)*100/S4
04080 PRINT ' WOMEN,MONTHLY INCIDENCE =',Q3,'CHANGE (%)= ',(Q3-S3)*100/S3
04100 PRINT ' MEN, MONTHLY INCIDENCE =',Q2,'CHANGE (%)= ',(Q2-S2)*100/S2
04105 IF R1=1 THEN 4200
04110 PRINT ' WOMEN,DISCOVERED BY POPULATION SCREENING=':C
04116 IF R1=2 THEN 4124
04120 PRINT ' WOMEN,DISCOVERED BY SUPPLEMENTARY CONTROL=':.1*C
04124 PRINT 'THE PROBABILITY THAT A WOMAN DISCOVERED BY POPULATION SCREENING'
04126 PRINT ' IS A GP1 MEMBER IS':R(1)
04130 IF R1>2 THEN 4136
04132 E2=Q2
04134 E3=Q3
04135 E7=Q5
04136 IF R1=2 THEN 4200
04138 PRINT 'THE PROBABILITY THAT A WOMAN DISCOVERED BY THE SUPPLEMENTARY'
04140 PRINT ' PROCEDURE IS A GP1 MEMBER IS':P(1)
04144 IF R1>3 THEN 4148
04146 C1=(E7-Q5)/E7*100
04148 C2=(E7-Q5)/E7*100
04150 PRINT 'IF THIS SUPPLEMENTARY CONTROL PROCEDURE INCREASES THE NUMBER OF'
04152 PRINT ' DISCOVERIES OF INFECTIOUS WOMEN BY 10%, THEN THE % DECREASE'
04154 PRINT ' IN PREVALENCE IN WOMEN IS':C2:'WHICH IS':C2/C1:'TIMES THE % '
04158 PRINT ' DECREASE IN PREVALENCE IN WOMEN DUE TO SUPPLEMENTARY POPULATIION'
04160 PRINT ' SCREENING.'

```

```

04170 E4=(E2-Q2)/(.1*C)
04172 E5=(E3-Q3)/(.1*C)
04174 PRINT 'THE DISCOVERY AND CURE OF ONE INFECTIOUS WOMAN BY THIS '
04176 PRINT '      SUPPLEMENTARY CONTROL PROCEDURE PREVENTS':E5:'CASES IN'
04178 PRINT '      WOMEN AND':E4:' CASES IN MEN.'
04200 PRINT '-----'
04210 IF R1<>1 THEN 4214
04212 PRINT
04214 IF R1<>3 THEN 4220
04215 PRINT
04220 R1=R1+1
04240 IF R1>5 THEN 4300
04280 GOTO 2180
04300 PRINT 'END OF PROGRAM'
04320 REM ***** END OF MAIN PROGRAM LOOP *****
04340 STOP
04360 REM * * * * * SUBROUTINES * * * * *
04380 REM
04400 REM #####
04410 REM
04420 REM          SOLVER SUBROUTINE
04440 REM THE PROCEDURE TO FIND PREVALENCES Y(I) BEGINS HERE
04460 REM
04480 REM J2 COUNTS THE NUMBER OF ITERATIONS MADE SO FAR
04500 REM J3 SELECTS THE METHOD:  J3=0 FOR EULER STEPS ALONG TRAJECTORY:
04520 REM                      J3=1 WHEN WE USE THE NEWTON METHOD
04540 REM J5=0 IF WE DON'T WISH TO COMPUTE DIFFERENCES IN THE DIRECTION FIELD
04560 J2=0
04580 J3=0
04600 J5=0
04780 IF G2=0 THEN 5000
04800 PRINT 'PREVALENCES, AS FRACTIONS OF EACH GROUP'
04820 PRINT 'CORE WOMEN', 'CORE MEN', 'NONCORE WOMEN', 'NONCORE MEN', 'RMS RATE'
04840 PRINT Y(1),Y(2),Y(3),Y(4)
04860 GOTO 5000
04880 REM THE ITERATION LOOP BEGINS HERE
04900 IF G2=0 THEN 5000
04920 PRINT Y(1),Y(2),Y(3),Y(4),W5
04940 REM
04960 REM DECIDE WHICH SOLUTION METHOD TO APPLY NEXT
04980 REM
05000 IF J3=1 THEN 5400
05020 IF J4=G THEN 5030
05025 GO TO 5160
05030 IF J2>9 THEN 5060
05040 GOTO 5160
05060 J3=1
05080 IF G2=0 THEN 5160
05100 PRINT 'NEWTON METHOD BEGINS AFTER NEXT STEP'
05120 REM THIS ITERATION PROCEDURE MAY BE USEFUL BEFORE NEWTON METHOD BEGINS:
05140 REM GO TO THE RATE SUBROUTINE TO USE THE MODEL EQUATIONS
05160 GOSUB 6880
05180 J5=1
05200 FOR I=1 TO G STEP 1
05220 REM Y(I)=(W(I)-S1*R(I))*D(I)/(1+D(I)*W(I)) IS AN ALTERNATIVE
05240 REM EULER STEPS ALONG TRAJECTORY:
05260 Y(I)=Y(I)+.25*V(I,1)
05280 IF Y(I)>0 THEN 5320
05300 Y(I)=0
05320 NEXT I

```

```

05340 GOTO 5720
05360 REM WE GO TO THE NEWTON SUBROUTINE AFTER SAVING THE OLD SOLUTION IN E
05380 REM AND THE OLD RMS RATE IN W7
05400 MAT E=Y
05420 W7=W5
05440 GOSUB 5820
05460 IF W5<W7 THEN 5720
05480 IF G2=0 THEN 5520
05500 PRINT 'USE ONLY HALF OF INCREMENT SINCE THE RMS RATE=',W5
05520 MAT Y=E+Y
05540 MAT Y=(.5)*Y
05560 GOSUB 5820
05580 IF W5<W7 THEN 5720
05600 MAT Y=E
05620 J3=0
05640 IF G2=0 THEN 5680
05660 PRINT 'USE STARTER AGAIN SINCE RMS RATE=',W5
05680 GOTO 5160
05700 REM STEP UP ITERATION COUNTER
05720 J2=J2+1
05740 IF W5>1E-06 THEN 5750
05745 RETURN
05750 IF J2<300 THEN 4900
05755 PRINT 'WARNING: 300 ITERATIONS'
05760 RETURN
05780 REM #####
05800 REM
05820 REM NEWTON SUBROUTINE STARTS HERE
05840 REM
05860 REM WE WILL SOLVE F(X)=0, WHERE X CONTAINS PREVALENCES AND F CONTAINS
05880 REM THEIR DIRECTION FIELD.
05900 REM
05920 FOR I=1 TO G STEP 1
05940 X(I,1)=Y(I)
05960 F(I,1)=V(I,1)
05980 NEXT I
06000 REM GO TO SUB TO FORM PARTIAL DERIVATIVE MATRIX A
06020 GOSUB 6420
06040 REM
06060 REM FIND NEW PREVALENCES X BY NEWTON METHOD, THEN STORE THEM IN Y
06080 REM
06100 MAT L=INV(A)
06120 MAT H=L*F
06140 MAT X=X-H
06160 REM
06180 FOR I=1 TO G STEP 1
06200 Y(I)=X(I,1)
06220 NEXT I
06240 GOSUB 6880
06260 RETURN
06280 REM NEWTON ALGORITHM ENDS HERE
06300 REM
06320 REM #####
06340 REM
06360 REM SUBROUTINE FDERIV: FORM APPROXIMATE PARTIAL DERIVATIVE MATRIX A
06380 REM
06400 REM
06420 H9=.001
06440 J9=1
06460 Y(J9)=Y(J9)-H9

```

```

06480 GOSUB 6880
06500 J5=1
06520 Y(J9)=Y(J9)+2*H9
06540 GOSUB 6880
06560 REM STORE PARTIALS: MATRIX A APPROXIMATES DF
06580 FOR I=1 TO G STEP 1
06600 A(I,J9)=V(I,3)/(2*H9)
06620 NEXT I
06640 Y(J9)=Y(J9)-H9
06660 J9=J9+1
06680 J5=0
06700 IF J9<=G THEN 6460
06720 RETURN
06740 REM
06760 REM #####
06780 REM
06800 REM ***** RATE SUBROUTINE *****
06820 REM
06840 REM WE COMPUTE THE DIRECTION FIELD V(I,1) USING THE MODEL EQUATIONS
06860 REM
06880 FOR I=1 TO G STEP 1
06920 W1=0
06940 FOR J1=1 TO G STEP 1
06960 W1=W1+B(I,J1)*Y(J1)
06980 NEXT J1
07000 W(1)=W1
07020 NEXT I
07040 REM
07060 REM GO TO SUB TO FORM EXTRA CURE RATES R(I)
07080 REM
07100 GOSUB 7580
07120 REM
07140 REM COMPUTE V(1,J), TIME RATES OF CHANGE OF PREVALENCES Y(I)
07160 REM
07180 W5=0
07200 J4=0
07220 FOR I=1 TO G STEP 1
07240 REM ***** MODEL EQUATIONS: *****
07260 V(1,1)=(1-Y(I))*W(I)-(Y(I)/D(I))-C*R(I)/N(I)-.1*C*P(I)/N(I)
07280 IF J5=0 THEN 7400
07300 REM CALCULATE CHANGES IN SLOPES
07320 V(1,3)=V(I,1)-V(I,2)
07340 IF J3=1 THEN 7400
07360 IF SGN(V(1,3))=SGN(V(I,1)) THEN 7400
07380 J4=J4+1
07400 V(I,2)=V(I,1)
07420 W5=W5+V(1,1)*V(I,1)
07440 NEXT I
07460 W5=SQR(W5)
07480 RETURN
07500 REM *****
07520 REM ***** CONTROL SUBROUTINE *****
07540 REM THESE LOOPS FORM EXTRA CURE RATES R(I) AND P(I)
07560 REM
07580 IF R1=1 THEN 7770
07586 IF R1>2 THEN 7720
07590 T1=C
07600 C=0
07620 FOR I=1 TO G-1 STEP 2
07640 Z(1)=Y(I)*N(J)*(1/D(I)+T1*R(I)/(Y(I)*N(I)))

```

```

07660 C=C+Z(I)
07680 NEXT I
07700 C=.1*C
07710 REM C IS THE DISCOVERY RATE DUE TO POPULATION SCREENING
07720 IF R1<4 THEN 7880
07740 IF R1=4 THEN 8080
07760 IF R1=5 THEN 8360
07770 REM UNCONTROLLED CASE
07780 FOR I=1 TO G STEP 1
07800 R(1)=0
07810 P(I)=0
07820 NEXT I
07860 RETURN
07880 REM POPULATION SAMPLING, WOMEN
07900 REM
07920 A3=0
07940 FOR I=1 TO G-1 STEP 2
07960 A3=A3+Y(1)*N(I)
07980 NEXT I
08000 FOR I=1 TO G-1 STEP 2
08020 R(I)=Y(I)*N(I)/A3
08040 NEXT I
08050 IF R1=2 THEN 8060
08052 FOR I=1 TO G-1 STEP 2
08054 P(I)=R(I)
08056 NEXT I
08060 RETURN
08080 REM INFECTEE SAMPLING, WOMEN
08100 A5=0
08120 FOR I=1 TO G-1 STEP 2
08140 A4=0
08160 FOR I9=2 TO G STEP 2
08180 A4=A4+M(1,I9)*(1-Y(I))*Y(I9)*N(I9)*K(I9)/D(I9)
08200 NEXT I9
08220 P(I)=A4
08240 A5=A5+P(I)
08260 NEXT I
08280 FOR I=1 TO G-1 STEP 2
08300 P(I)=P(I)/A5
08320 NEXT I
08340 RETURN
08360 REM INFECTOR SAMPLING, WOMEN
08380 A5=0
08400 FOR I=1 TO G-1 STEP 2
08420 A4=0
08440 FOR I9=2 TO G STEP 2
08460 A4=A4+M(I9,I)*(1-Y(I9))
08480 NEXT I9
08500 P(I)=A4*N(I)*Y(I)*K(I)/D(I)
08520 A5=A5+P(I)
08540 NEXT I
08560 FOR I=1 TO G-1 STEP 2
08580 P(I)=P(I)/A5
08600 NEXT I
08620 RETURN
08640 REM #####
08660 REM
08680 MAT M=ZER(G,G)
08700 MAT T=ZER(G,G)
08720 MAT B=ZER(G,G)

```



```

08760 REM SET UP PROPORTIONATE MIXING MODEL
08780 A1=0
08800 A2=0
08820 FOR I=1 TO G STEP 1
08840 U(I)=N(I)*K(I)/D(I)
08860 NEXT I
08880 FOR I=2 TO G STEP 2
08900 A1=A1+U(I-1)
08920 A2=A2+U(I)
08940 NEXT I
08960 FOR I9=1 TO G-1 STEP 2
08980 FOR I=2 TO G STEP 2
09000 M(I-1,I9+1)=U(I-1)/A1
09020 M(I,I9)=U(I)/A2
09040 NEXT I
09060 NEXT I9
09080 IF G1=0 THEN 9440
09120 MAT L=ZER(G,G)
09130 FOR K8=3 TO 7 STEP 4
09140 FOR K9=1 TO 2
09150 U8=U(K9)+U(K9+4)
09160 L(K9,K8-K9)=U(K9)/U8
09170 L(K9+4,K8-K9)=U(K9+4)/U8
09180 U9=U(K9+2)+U(K9+6)
09190 L(K9+2,K8-K9+2)=U(K9+2)/U9
09200 L(K9+6,K8-K9+2)=U(K9+6)/U9
09210 NEXT K9
09220 NEXT K8
09250 MAT L=(G1)*L
09260 MAT M=(1-G1)*M
09270 MAT M=M+L
09440 FOR I9=1 TO G STEP 1
09460 FOR I=1 TO G STEP 1
09480 T(I,I9)=M(I,I9)*K(I9)
09500 B(I,I9)=T(I,I9)*N(I9)/(D(I9)*N(I))
09520 NEXT I
09540 NEXT I9
09745 IF G1>0 THEN 9780
09748 PRINT
09750 PRINT 'PROB. OF TRANS. BY INF. MAN/PROB. OF TRANS. BY INF. WOMAN=':R7
09752 PRINT 'SEXUAL ACTIVITY OF GP1 WOMAN/SEXUAL ACT. OF GP3 WOMAN=':R8
09754 PRINT 'SEXUAL ACTIVITY OF GP2 MAN/SEXUAL ACTIVITY OF GP4 MAN=':R9
09756 P1=0
09758 P2=0
09760 FOR I=2 TO G STEP 2
09762 P1=P1+M(I,1)*K(I)
09764 P2=P2+M(I-1,2)*K(I-1)
09766 NEXT I
09768 K4=P1*P2
09770 PRINT'EFFECTIVE CONTACT NUMBER=':K4
09772 PRINT
09780 IF G2=0 THEN 9880
09800 PRINT 'MIXING MATRIX: M(I,J)=FRAC OF EFF. CONTACTS OF J THAT ARE WITH I'
09840 PRINT 'CONTACT MATRIX B'
09880 RETURN
09900 END

```